

```

1 *****
2 *
3 *      File and Aux Typer - An AppleWorks Init      *
4 *      *****                                     *
5 *      - Expands the 'Change file type'            *
6 *      selection in the File Activities            *
7 *      menu of AppleWorks 5.1 by:                 *
8 *      (i) adding several menu-selectable         *
9 *      popular file types from which              *
10 *      to chose                                   *
11 *      (ii) displaying both the current file      *
12 *      type and the aux type of the               *
13 *      selected file                              *
14 *      (iii) displaying both the 3-character      *
15 *      file type (e.g. AWP) and its              *
16 *      hexadecimal representation                *
17 *      (e.g. $1A)                                 *
18 *      (iv) allowing entry of a file's           *
19 *      type and aux type in hexadecimal          *
20 *      (e.g. $FC / $8001)                        *
21 *
22 *      Version 5.1 (for AppleWorks Version 5.1)   *
23 *      (c) 2024 by Hugh Hood                     *
24 *
25 *****
26 *
27 *      - Seg $2A (Disk/File activities) is patched to *
28 *      add the new/changed features to the File   *
29 *      Activities menu                             *
30 *
31 *      - NOTE: Since the length of the new code for $8200+ *
32 *      exceeds the $400 byte maximum for the     *
33 *      SEG Patch Manager memory ($BB00-$BEFF),   *
34 *      the code is first stored in Desktop       *
35 *      memory when the init is run and its      *
36 *      pointer is saved within the patch code    *
37 *      so that it may be retrieved and run at   *
38 *      the proper address ($8200+).             *
39 *
40 *****
41
42      TR          ADR          ; truncate bank address
43
44      XC          ; enable 65C02 code
45
46
47 * Internal AppleWorks Equates
48 ErrFlag      EQU          $03          ; ProDOS error occur?
49
50 FoundCmd     EQU          $86          ;
51 FoundRtn     EQU          $87          ;
52 FoundEsc     EQU          $88          ; >0 if <esc> pressed
53 MReg         EQU          $91          ; holds result of math operations
54 AArg         EQU          $9A          ; used for ZP pointers
55 CurrMode     EQU          $CC7        ; top center msg
56 StackP1      EQU          $CDC        ; 1st routine in stack (top right)
57 NewStr       EQU          $D85        ; 128 bytes for string storage
58 OldStr       EQU          $E05        ; 128 bytes for string storage

```

```

59
60 * AppleWorks Host Equates
61 AVersion      EQU          $1003      ; $33/51 = 5.1 / $28/40 = 4.0 /
62                                     ; $1E/30 = 3.0
63 AfterReadTest EQU          $100C      ; JSR here after getting key
64                                     ; (normally $60/RTS)
65 TOActive      EQU          $10E6      ; $00 = enabled / $01 = disabled <0A-ESC>
66 DoBell        EQU          $1115      ;
67 GetVBar       EQU          $1130      ; returns selection from vertical menu
68 MvLeftRtn     EQU          $1148      ; move in memory routine (negative/left)
69                                     ; - followed by destination/source/length
70 MvRightRtn    EQU          $114B      ; move in memory routine (positive/right)
71                                     ; - followed by destination/source/length
72 PopStack      EQU          $114E      ; removes last escape level
73 PushStack     EQU          $1157      ; add level to escape menu
74 RestCursor    EQU          $115D      ; restore cursor position
75 SetUCC        EQU          $1163      ; convert accum char a-z to uppercase
76 StrMvRtn      EQU          $116C      ; follow with to/from
77 StrWrRtn      EQU          $116F      ; follow with column/row/string address
78 WriteCom      EQU          $118A      ; writes a string on command line
79 Write?        EQU          $118D      ; are you sure?
80
81 * AppleWorks SEG $2A (File and Disk Activities) Equates
82
83
84
85 NewCodeAdr     EQU          $8200      ; new code to be placed here
86                                     ; (SEG $2A ends at $8100)
87
88 * AppleWorks Subhost Equates
89 WriteText     EQU          $AB37      ; writes multiple strings
90                                     ; - follow with address of string 'block'
91                                     ; - block contains: (i) pString for 1
92                                     ; (ii) column for 1
93                                     ; (iii) row for 1
94                                     ; (iv) repeat i/ii/iii
95                                     ; (v) $00 = endbyte
96 ManyVBar     EQU          $AB3A      ; routine to build a vertical menu
97                                     ; (repeatedly calls VBarWrRtn)
98                                     ; - followed by address of string 'block'
99                                     ; - then: Column for #1 (+$80 if work area)
100                                    ; Row for #1 (+$80 if work area)
101                                    ; Spacing (e.g. 1/2/3)
102                                    ; end byte ($00)
103                                     ; - block contains: (i) pString for 1
104                                     ; (ii) pString for 2
105                                     ; (iii) pString for n
106                                     ; (iv) $00 = endbyte
107 newMultByte   EQU          $AB70      ; Multiply X x Y {result to MReg - $91/$91}
108
109
110
111 SMGetBlock     EQU          $D005      ; load a block into main memory
112                                     ; (pointer in main / 'gets' to main)
113                                     ; - follow with:
114                                     ; (i) address in main to block pointer
115                                     ; (ii) address in main to write data
116                                     ; [CArg/ZReg00 $9E/$9F is returned with

```

```

117                                     ; number of bytes retrieved]
118 SMPutBlock      EQU          $D011      ; puts a block into desktop memory from main
119                                     ; - follow with:
120                                     ; (i) address in main to block pointer
121                                     ; [if possible, uses existing pointer;
122                                     ; otherwise new pointer stored at
123                                     ; this address]
124                                     ; (ii) address in main to read data
125                                     ; (iii) length of area to read
126
127 DispFE         EQU          $D02C      ; display function and escape map
128 GetStr         EQU          $D038      ; LDA with maxlength before calling
129 GetYN         EQU          $D03B      ; menu bar to get a yes or no
130
131 * AppleWorks Init Manager Equates
132 imSavePatch    EQU          $3006      ; Patch Manager save routine in SEG.IM
133 InitAdr        EQU          $4000      ; load address for Init files
134 PatchAdr       EQU          $BB00      ; initial load address for patch code
135                                     ; (NOTE: uses ProDOS I/O buffer -
136                                     ; 1K max length -
137                                     ; $BB00 - $BEFF)
138
139
140 * Literals
141
142 * BaseColumn   EQU          #$06      ; example only - not used here
143
144
145 *****
146
147             ORG          InitAdr      ; ($4000)
148             TYP          $06          ; create binary file
149
150
151 *****
152 *           Init Header           *
153 *****
154 START
155             JMP          IStart      ; skip over header
156
157 **-----
158
159             ASC          'mb'        ; Init ID Bytes (AW 5.1)
160             DB          $33          ; Init Version - programmer assigned
161                                     ; e.g. - $0A/1.0 $0B/1.1 $33/5.1
162             STR          'FileAuxTyper' ; Init Screen Name
163
164             HEX          0000        ; Init Header End Bytes
165
166 **-----
167
168 IStart
169
170             LDA          AWVersion    ; AppleWorks version #
171             CMP          #$33        ; Is it Version 5.1?
172             BNE         Done         ; disregard - wrong version
173
174

```

```

175 StoreStrings    JSR          SMPutBlock    ; store new code in DT memory
176                DA          NewCodePtrAdr ; save pointer here for retrieval ($4xxx)
177                DA          Code1End      ; start of new code to store ($4xxx)
178                DA          NewCodeEnd-Code1End ; length of new code to store
179
180
181 PatchH2C        JSR          imSavePatch ; call patch manager
182                DW          Code1         ; beginning of patch1 code ($40xx)
183                DW          Code1End-Code1 ; length of patch code
184                DW          $002A        ; SEG number to patch
185                ; ($2A = File and Disk Activities SEG)
186
187 Done            RTS          ; back to Init Manager
188
189 **-----
190
191 Code1           EQU          *          ; (will be $40xx)
192
193
194 * Begin $BB00+ run location addressing
195                ORG          PatchAdr    ; (Patching Code is moved and run
196                ; @ $BB00 by Init Manager)
197
198                HEX          0000       ; Init begin bytes for SEG $2A Patch
199
200
201                JSR          SMGetBlock  ; retrieve new code from DT memory
202                DA          NewCodePA    ; address of pointer stored by SMPutBlock
203                ; at running address ($BBxx+)
204                DA          NewCodeAdr   ; place New Code at $8200
205
206 * Write any new code and tables at end rather than moving existing code and tables
207
208                LDA          #$20       ; JSR
209                STA          $6E22      ;
210                LDA          #<NewMenu ;
211                STA          $6E23      ;
212                LDA          #>NewMenu ;
213                STA          $6E24      ;
214
215
216                LDA          #$0B       ; new row #
217                STA          $6E1F      ;
218                LDA          #<Prompt1  ;
219                STA          $6E20      ;
220                LDA          #>Prompt1  ;
221                STA          $6E21      ;
222
223
224 * Modify File activities selection to show 'Change file and aux types'
225                LDA          #<Block5   ;
226                STA          $6318      ;
227                LDA          #>Block5   ;
228                STA          $6319      ;
229
230 * Modify command line prompt message to show $ *
231
232                LDA          #<CmdStr1  ;

```

```

233         STA             $6E46             ;
234         LDA             #>CmdStr1         ;
235         STA             $6E47             ;
236
237 * Modify Max Number Used for GetNum ($FF/255) to Max Length used by GetStr
238         LDA             #$02
239         STA             $6E49
240
241 * Modify GetNum to use OurGetStr Instead
242         LDA             #<OurGetStr       ;
243         STA             $6E4B             ;
244         LDA             #>OurGetStr       ;
245         STA             $6E4C             ;
246
247
248 * select ProDOS hex file type for menu selection chosen using NewTable1
249
250         LDA             #<NewTable1-1    ;
251         STA             $6E32             ;
252         STA             $6E60             ;
253         LDA             #>NewTable1-1    ;
254         STA             $6E33             ;
255         STA             $6E61             ;
256
257 * increase # of items to check in larger NewTable1
258
259         LDA             #$17             ; 22 items + 1
260         STA             $6E66             ; 10 items + 1
261
262 * experimental: redirect ESC at enter custom back to selection menu
263         LDA             #$4C             ; JMP
264         STA             $6E5A             ;
265         LDA             #$22             ;
266         STA             $6E5B             ;
267         LDA             #$6E             ;
268         STA             $6E5C             ;
269
270 * to replace routine lost when $6E5A was patched to return
271 * to NewMenu when <ESC> is pressed in enter Custom File option
272         LDA             #$4C             ; JMP
273         STA             $6E2B             ;
274         LDA             #<MyGetVBar      ;
275         STA             $6E2C             ;
276         LDA             #>MyGetVBar      ;
277         STA             $6E2D             ;
278
279 * experimental: change JMP after ESC address from $6D8A to $6D8E to
280 * prevent double count of ESC
281 * {Seems to be AppleWorks bug that also effects other options}
282         LDA             #$8E             ;
283         STA             $6E16             ;
284
285 * experimental: require confirmation before changing type
286         LDA             #$4C             ; JMP
287         STA             $6E39             ; was LDA #$01
288         LDA             #<MyConfirmRtn   ;
289         STA             $6E3A             ;
290         LDA             #>MyConfirmRtn   ;

```

```

291          STA          $6E3B          ; was RTS
292
293          RTS
294
295 *-----
296
297
298 PatchEnd    EQU      *              ; ($BBxx=+)
299
300
301 *-----
302
303          ORG
304 NewCodePtrAdr          ; ($4xxx+)
305          ; pointer stored here by SMPutBlock above
306          ORG      PatchEnd          ; return to $BBxx addressing
307
308 NewCodePA    DA      $0000          ; pointer to stored code block ($BBxx+)
309
310 **-----
311
312          ORG
313 Code1End
314
315 **-----
316
317          ORG      NewCodeAdr        ; $8200
318
319 **-----
320 NewCode
321
322 NewMenu
323          STZ      ChangeFlag        ; initialize ChangeFlag
324
325
326 NewMenu1
327          JSR      ManyVBar          ;
328          DA      Block0              ;
329          DB      $10                  ; column
330          DB      $0D                  ; row
331          DB      $01                  ; spacing
332          DB      $00                  ; end byte
333
334          JSR      ManyVBar          ;
335          DA      Block1              ;
336          DB      $10                  ; column
337          DB      $10                  ; row
338          DB      $01                  ; spacing
339          DB      $00                  ; end byte
340
341          JSR      ManyVBar          ;
342          DA      Block2              ;
343          DB      $1F                  ; column
344          DB      $10                  ; row
345          DB      $01                  ; spacing
346          DB      $00                  ; end byte
347
348          JSR      ManyVBar          ;

```

```

349      DA      Block3      ;
350      DB      $2E          ; column
351      DB      $10          ; row
352      DB      $01          ; spacing
353      DB      $00          ; end byte
354
355      JSR      ManyVBar    ;
356      DA      Block4      ;
357      DB      $3D          ; column
358      DB      $10          ; row
359      DB      $01          ; spacing
360      DB      $00          ; end byte
361
362      LDA      ChangeFlag   ;
363      BEQ     :NM2         ; nothing changed yet / 1st time through
364
365      RTS                                     ; goes back to $6E25 {one after JSR}
366
367
368 :NM2      JSR      StrWrRtn ;
369      DB      $30          ; column
370      DB      $0D          ; row
371      DA      Line1and2    ;
372
373      JSR      StrWrRtn    ;
374      DB      $30          ; column
375      DB      $0E          ; row
376      DA      Line1and2    ;
377
378
379 * Read hex bytes in File Info parm table and show in ASCII on screen
380
381      LDA      $807A        ; file type from parm table
382
383      JSR      Hex2Asc      ; convert to ASCII 0-9/A-F
384      STX     CrntTypeStr+11 ;
385      STA     CrntTypeStr+12 ;
386
387      JSR      HexTo3Char   ; match hex type to 3-character code
388
389      JSR      StrWrRtn    ;
390      DB      $34          ; column
391      DB      $0D          ; row
392      DA      CrntTypeStr  ;
393
394      JSR      StrWrRtn    ;
395      DB      $40          ; column
396      DB      $0D          ; row
397      DA      TypeStr      ;
398
399      LDA      $807C        ; aux type (HB) from parm table
400      JSR      Hex2Asc      ; convert to ASCII 0-9/A-F
401      STX     CrntAuxStr+11 ;
402      STA     CrntAuxStr+12 ;
403
404      LDA      $807B        ; aux type (LB) from parm table
405      JSR      Hex2Asc      ; convert to ASCII 0-9/A-F
406      STX     CrntAuxStr+13 ;

```

```

407          STA          CrntAuxStr+14 ;
408
409          JSR          StrWrRtn      ;
410          DB          $34           ; column
411          DB          $0E           ; row
412          DA          CrntAuxStr    ;
413
414 ***
415
416          JSR          StrWrRtn      ;
417          DB          $31           ; column
418          DB          $0C           ; row
419          DA          TopLine       ;
420
421          JSR          StrWrRtn      ;
422          DB          $31           ; column
423          DB          $0F           ; rown
424          DA          BottomLine    ;
425
426          RTS          ; goes back to $6E25 {one after JSR}
427
428 *          JMP          $6E25      ; back to regular code
429
430 Block0     STR          'Enter new hex file type' ;
431           STR          'Enter new hex aux type'  ;
432           DB          $00          ; end byte
433
434 Block1     STR          '$19/ADB'          ;
435           STR          '$1A/AWP'          ;
436           STR          '$1B/ASP'          ;
437           STR          '$04/TXT'          ;
438           STR          '$06/BIN'          ;
439           DB          $00          ; end byte
440
441 Block2     STR          '$C8/FON'          ;
442           STR          '$FC/BAS'          ;
443           STR          '$FF/SYS'          ;
444           STR          '$50/GWP'          ;
445           STR          '$51/GSS'          ;
446           DB          $00          ; end byte
447
448 Block3     STR          '$52/GDB'          ;
449           *          STR          '$16/$B3'   ;
450           *          STR          'NDA/$B8'   ;
451           STR          '$07/FNT'          ;
452           STR          '$08/FOT'          ;
453           STR          '$B0/SRC'          ;
454           STR          '$E0/LBR'          ;
455           DB          $00          ; end byte
456
457 Block4     STR          '$B9/CDA'          ;
458           STR          '$C1/PIC'          ;
459           STR          '$CA/ICN'          ;
460           STR          '$F9/OS '          ;

```



```

465 STR '$0F/DIR' ;
466 DB $00 ; end byte
467
468 Prompt1 STR 'Enter new file/aux type or select:'
469
470 CrntTypeStr STR 'Current: $ ' ;
471 CrntAuxStr STR 'Current: $ ' ;
472 PendingStr STR 'Pending'
473 PendTypeStr STR 'Pending: $ ' ;
474 PendAuxStr STR 'Pending: $ ' ;
475 TypeStr STR '/XXX' ;
476
477 TopLine STR ' _____ ' ; Underscore for Top Line
478
479 BottomLine STR "LLLLLLLLLLLLLLLLLLLLLLLLLLLL" ; MouseText 'L' is Top Bar
480
481 Line1and2 DB :L12-*--1 ; Leading Length Byte
482 ASC "Z" ; MouseText Right Bar
483 ASC ' '
484 ASC "_ " ; MouseText Left Bar
485 :L12
486
487 CmdStr1 STR 'Enter new 2-digit hex file type: $'
488 CmdStr2 STR 'Enter new 4-digit hex aux type: $'
489
490 AnotherStr STR 'Make another change to this file'
491
492 ChangeFlag DB #$00 ; not zero if change made
493
494 FileNameStr DS 16 ; temp space for stack move
495
496 **-----
497 * Replacement menu block for $5CE7
498
499 Block5
500 STR 'Lock files' ;
501 STR 'Unlock files' ;
502 B53 STR 'Change file and aux types' ;
503 DB $00 ; end byte
504
505 * ChangeTitle STR 'Change file/aux types' ; note too long for stack
506 ; just use default
507
508
509 **-----
510 * New file type table for $6222
511
512 NewTable1
513 DB $00 ; enter new file type
514 DB $00 ; enter new aux type
515 DB $19 ; ADB
516 DB $1A ; AWP
517 DB $1B ; ASP
518 DB $04 ; TXT
519 DB $06 ; BIN
520 DB $C8 ; FON
521 DB $FC ; BAS
522 DB $FF ; SYS

```

```

523          DB          $50          ; GWP
524          DB          $51          ; GSS
525          DB          $52          ; GDB
526 *        DB          $B3          ; S16
527 *        DB          $B8          ; NDA
528          DB          $07          ; FNT
529          DB          $08          ; FOT
530          DB          $B0          ; SRC
531          DB          $E0          ; LBR
532          DB          $B9          ; CDA
533          DB          $C1          ; PIC
534          DB          $CA          ; ICN
535          DB          $F9          ; OS
536          DB          $0F          ; DIR
537
538 **-----
539
540 * Hex2Asc Routine - Input 'A' and Output 'X' (HNib) and 'A' (LNib)
541
542 Hex2Asc
543          PHA          ;
544          LSR          ;
545          LSR          ;
546          LSR          ;
547          LSR          ;
548          JSR          :A          ;
549          TAX          ;
550          PLA          ;
551          AND          #$0F          ;
552 :A        ORA          #$30          ;
553          CMP          #$3A          ;
554          BCC          :B          ;
555          ADC          #$06          ;
556 :B        RTS          ;
557
558 **-----
559
560
561 * Patch GetStr to just get 0-9 and A-F before calling GetStr
562 * and then restore GetStr to default when done
563
564 OurGetStr
565          STA          StrSize      ; save entry accumulator / 2 or 4
566          STZ          FoundEsc     ; clear ESC on 2nd trip through
567
568 :PP      INC          T0Active     ; disable TimeOut until we un-patch
569          ; after GetStr
570          LDA          #$4C          ; JMP
571          STA          $B245        ;
572          LDA          #<HexFilter  ;
573          STA          $B245+1      ;
574          LDA          #>HexFilter  ;
575          STA          $B245+2      ;
576
577          STZ          NewStr       ; init Newstr
578
579          LDA          StrSize      ; 2 or 4 characters
580          JSR          GetStr       ;

```





```

697 :ZZ          LDA          #$00          ; ESC pressed;
698          RTS          ; go back next file and/or
699          ; to File Activities menu
700
701 :YY          JMP          $6E3C         ; Enter Custom File Type Routine
702
703 * Change aux type routine - duplicate (with changes) routine @ $6E3C
704 :XX          JSR          StrMvRtn      ; preserve OldStr
705          ; (contains pathname of selected file)
706          DA          $BB00          ; temp storage area
707          DA          OldStr         ; $0E05
708          JSR          WriteCom       ;
709          DA          CmdStr2        ; enter new 4-digit hex aux type: $
710          LDA          #$04          ; max # of hex digits allowed
711          JSR          OurGetStr
712
713          JSR          StrMvRtn      ; restore OldStr
714          ; (contains pathname of selected file)
715          DA          OldStr         ; restore OldStr ($E05) [file pathname]
716          DA          $BB00          ;
717
718 * check for <esc> pressed
719          LDX          FoundEsc       ;
720          BNE          :XXZ          ;
721
722 * no <esc> - take two aux bytes and put in parm table
723          LDA          HexByte2       ;
724          STA          $807B         ; SetFileInfo Parms Aux Type (LB)
725          LDA          HexByte1       ;
726          STA          $807C         ; SetFileInfo Parms Aux Type (HB)
727
728          JMP          $6E39         ; LDA #$01 (success) and RTS to next file
729          ; and/or to File Activities menu
730
731 :XXZ         LDA          #$00          ; Escape pressed in entry; try again
732          JMP          $6E22         ; re-build menu and try again
733
734 **-----
735
736 * Read hex bytes in File Info parm table and show in ASCII on screen and confirm
737
738 MyConfirmRtn LDA          $807A         ; file type from parm table
739          JSR          Hex2Asc        ; convert to ASCII 0-9/A-F
740          STX          PendTypeStr+11 ;
741          STA          PendTypeStr+12 ;
742
743          JSR          HexTo3Char     ; match hex type to 3-character code
744
745          JSR          StrWrRtn       ;
746          DB          $34            ; column
747          DB          $0D            ; row
748          DA          PendTypeStr     ;
749
750          JSR          StrWrRtn       ;
751          DB          $40            ; column
752          DB          $0D            ; row
753          DA          TypeStr        ;
754

```



```

813          CLC                                ; prepare to add to start of 3CharTable
814          LDA                                #<3CharTable-1 ; (LB)
815          ADC                                MReg          ; (LB) / $91
816          TAX                                ;
817          LDA                                #>3CharTable  ; (HB)
818          ADC                                MReg+1        ; (HB) / $92
819          BRA                                :CCC          ; will always branch
820
821 * If no 3-character File Type Code, just display '$' plus Hex Bytes
822 :No3Code   JSR                                Hex2Asc      ; enter with File Type in 'A'
823          STX                                TypeStr+3     ; high nibble
824          STA                                TypeStr+4     ; low nibble
825          LDA                                #'$'         ; Hex symbol
826          STA                                TypeStr+2     ;
827
828          RTS
829
830 :CCC       STX                                AArg         ; (LB) of 3CharTable / $9A
831          STA                                AArg+1       ; (HB) of 3CharTable / $9B
832          LDY                                #$03         ; length of character string
833
834 :DDD       LDA                                (AArg),Y    ;
835          STA                                TypeStr+1,Y  ;
836          DEY                                ;
837          BNE                                :DDD         ; loop to get all 3 characters
838
839          RTS
840
841 HexTable
842          DB                                $00
843          DB                                $01
844          DB                                $02
845          DB                                $03
846          DB                                $04
847          DB                                $05
848          DB                                $06
849          DB                                $07
850          DB                                $08
851          DB                                $09
852          DB                                $0A
853          DB                                $0B
854          DB                                $0C
855          DB                                $0F
856          DB                                $10
857          DB                                $11
858          DB                                $12
859          DB                                $13
860          DB                                $14
861          DB                                $15
862          DB                                $16
863          DB                                $19
864          DB                                $1A
865          DB                                $1B
866          DB                                $20
867          DB                                $2A
868          DB                                $2B
869          DB                                $2C
870          DB                                $2D

```

871	DB	\$2E
872	DB	\$40
873	DB	\$41
874	DB	\$42
875	DB	\$50
876	DB	\$51
877	DB	\$52
878	DB	\$53
879	DB	\$54
880	DB	\$55
881	DB	\$56
882	DB	\$57
883	DB	\$58
884	DB	\$59
885	DB	\$5A
886	DB	\$5B
887	DB	\$5C
888	DB	\$5D
889	DB	\$5E
890	DB	\$5F
891	DB	\$6B
892	DB	\$6D
893	DB	\$6E
894	DB	\$6F
895	DB	\$A0
896	DB	\$AB
897	DB	\$AC
898	DB	\$AD
899	DB	\$B0
900	DB	\$B1
901	DB	\$B2
902	DB	\$B3
903	DB	\$B4
904	DB	\$B5
905	DB	\$B6
906	DB	\$B7
907	DB	\$B8
908	DB	\$B9
909	DB	\$BA
910	DB	\$BB
911	DB	\$BC
912	DB	\$BD
913	DB	\$BF
914	DB	\$C0
915	DB	\$C1
916	DB	\$C2
917	DB	\$C3
918	DB	\$C5
919	DB	\$C6
920	DB	\$C7
921	DB	\$C8
922	DB	\$C9
923	DB	\$CA
924	DB	\$D5
925	DB	\$D6
926	DB	\$D7
927	DB	\$D8
928	DB	\$DB



929	DB	\$E0
930	DB	\$E2
931	DB	\$EE
932	DB	\$EF
933	DB	\$F0
934	DB	\$F9
935	DB	\$FA
936	DB	\$FB
937	DB	\$FC
938	DB	\$FD
939	DB	\$FE
940	DB	\$FF

941  
942 3CharTable

943	ASC	'NON'
944	ASC	'BAD'
945	ASC	'PCD'
946	ASC	'PTX'
947	ASC	'TXT'
948	ASC	'PDA'
949	ASC	'BIN'
950	ASC	'FNT'
951	ASC	'FOT'
952	ASC	'BA3'
953	ASC	'DA3'
954	ASC	'WPF'
955	ASC	'SOS'
956	ASC	'DIR'
957	ASC	'RPD'
958	ASC	'RPI'
959	ASC	'AFD'
960	ASC	'AFM'
961	ASC	'AFR'
962	ASC	'SCL'
963	ASC	'PFS'
964	ASC	'ADB'
965	ASC	'AWP'
966	ASC	'ASP'
967	ASC	'TDM'
968	ASC	'8SC'
969	ASC	'8OB'
970	ASC	'8IC'
971	ASC	'8LD'
972	ASC	'P8C'
973	ASC	'DIC'
974	ASC	'OCR'
975	ASC	'FTD'
976	ASC	'GWP'
977	ASC	'GSS'
978	ASC	'GDB'
979	ASC	'DRW'
980	ASC	'GDP'
981	ASC	'HMD'
982	ASC	'EDU'
983	ASC	'STN'
984	ASC	'HLP'
985	ASC	'COM'
986	ASC	'CFG'

987	ASC	'ANM'
988	ASC	'MUM'
989	ASC	'ENT'
990	ASC	'DVU'
991	ASC	'FIN'
992	ASC	'BIO'
993	ASC	'TDR'
994	ASC	'PRE'
995	ASC	'HDV'
996	ASC	'WP'
997	ASC	'GSB'
998	ASC	'TDF'
999	ASC	'BDF'
1000	ASC	'SRC'
1001	ASC	'OBJ'
1002	ASC	'LIB'
1003	ASC	'S16'
1004	ASC	'RTL'
1005	ASC	'EXE'
1006	ASC	'PIF'
1007	ASC	'TIF'
1008	ASC	'NDA'
1009	ASC	'CDA'
1010	ASC	'TOL'
1011	ASC	'DVR'
1012	ASC	'LDF'
1013	ASC	'FST'
1014	ASC	'DOC'
1015	ASC	'PNT'
1016	ASC	'PIC'
1017	ASC	'ANI'
1018	ASC	'PAL'
1019	ASC	'OOG'
1020	ASC	'SCR'
1021	ASC	'CDV'
1022	ASC	'FON'
1023	ASC	'FND'
1024	ASC	'ICN'
1025	ASC	'MUS'
1026	ASC	'INS'
1027	ASC	'MDI'
1028	ASC	'SND'
1029	ASC	'DBM'
1030	ASC	'LBR'
1031	ASC	'ATK'
1032	ASC	'R16'
1033	ASC	'PAS'
1034	ASC	'CMD'
1035	ASC	'OS'
1036	ASC	'INT'
1037	ASC	'IVR'
1038	ASC	'BAS'
1039	ASC	'VAR'
1040	ASC	'REL'
1041	ASC	'SYS'
1042	DB	\$00
1043		
1044		

```
1045 **-----
1046          ORG          ; ($4xxx+)
1047
1048 NewCodeEnd EQU      *          ; ($4xxx+)
1049
1050 **-----
1051
1052          SAV          I.FILEAUXTYPE ;
1053          LST          OFF
1054
1055          END
1056
1057 *=====
1058
1059
```